
JP Module 
 

       
 

JP Module is a MMC/SD card data logger and consumes low power. It allows creating, 
appending and reading data to files stored on MMC/SD media card. The files are stored in FAT16 
format and can be directly read using a PC with an MMC/SD card reader. JP Module also can 
format MMC/SD card with FAT16. 
 
SPECS 
 
Power:  5V VDC  <25mA 
Size:  42mm x 40mm (1.65” x 1.57”) 
Speed:  2400, 4800, 9600 or 19200 Baud. 
Media:  Uses MMC or SD memory cards from 32 MB to 2GB. 
 
PIN FUNCTIONS 
 
RX: Serial input connection to JP Module. The module allows 2400, 4800, 9600 or 

19200 baud based on J1 and J2 settings. 
 
TX: Serial output connection to JP Module. The module allows 2400, 4800, 9600 or 

19200 baud based on J1 and J2 settings. 
 
                        J1 and J2 Settings: 

     J1       J2      Baud 
      -        -      2400 
      x        -      4800 
      -       x      9600 
      x       x     19200 

- = connected 
x = disconnected 
 

GND: Power supply and serial ground. This MUST also be connected to ground on the 
device to allow the serial data to be sent to the module. 

 
5V:  Supply voltage to module. Vin may be 5.0V (±0.5V), with 25 milliamps of 

current. 



 
State: This pin shows the module work state. If the module is working fine, the state 

output level will be high (Active light will be on) If the module stops working for 
any reason, the state pin output level will be low (Active light will be off). 

 
Reset: This pin can be set at the low level to restart module, if the module stops working 

for any reason. 
 
SERIAL DATA FORMAT 
 

The serial data format is eight data bits, no parity and 1 stop bit (8N1). Characters are 
sent using standard ASCII values. Baud rate may be 2400, 4800, 9600 and 19200 depending on 
the settings of J1 and J2. 
 
COMMAND FORMAT 
 
 JP Module can be easily used with some simple commands to create a new file, write a 
date and save data to MMC/SD card. 

1) Input operates command: MC! 
a) MC = create a new file and write data to MMC/SD card. 
b) MW = open an existing file and clear it, then write new data to it. 
c) MA = open an existing file and append data to this file. 
d) MD = delete an existing file. 
e) MR = read data from an existing file. 
f) MF = format MMC/SD card with FAT16. 
g) MI = obtain file information. 

2) Key in file name: XXXXXXXX.XXX! ( FILENAME.TXT or FILENAME.CSV) 
3) Write a date on file: yyyymmddhhmmss!  ( 20070529123030 means 05/29/2007 12:30:30) 

(Note: 1) the command must be 2 characters in uppercase! 
           2) The file name must be 8 characters in uppercase or number.  
           3) Ext. name must be 3 characters in uppercase. 
           4) “!” is command end of character) 
 
Then you can start to input your data to MMC/SD card. The module can write 1 – 128 

serial data to MMC/SD card every time, for example:  
1234.11, 2345.45, 3456.33, 5678.99!!  
2222.03, 3333.22, 4444.08, 2399.44!! 
…… 
Data will be automatically written to MMC/SD card.  
(Note:  The module will send data to MMC/SD card after it gets “!!”.  It needs a pause to 
process those characters without missing the next data. So the module will turn state pin 
in “low”, then turn it in “high” after finishing the process. ) 
 
If you want to close a file and open another file, just input “##” in your last data., for 

example:   
3489.12, 4586.07, 2785.37, 5475.75##.  
(Note: The module will close the file and wait to open another file when it gets “##” 
characters.) 
 

FORMAT MMC/SD CARD: 
 JP Module can format MMC/SD card with FAT16.  



1) Input operates command: MF! 
2) Input volume label: XXXXXXXXXXX!! (11 characters in length. If less than 11 

characters are provided, the label will be padded with spaces. If null string is passed 
volume will not labeled.) 

(Note: When the module formats MMC/SD card, the state pin will be turned to “low” and 
it will turn to “high” after it finishes formatting.  
WARNING: Formatting will erase ALL data on your MMC/SD card!) 

 
MEDIA CARD SETUP: 
 

The file system must be FAT16 (just FAT under windows) format. You need a card 
reader to format the card. Insert the card in the reader then go to My Computer and right click the 
drive which represents your card. Then choose "Format". Make sure that you do not format your 
HDD! In format options dialog under "File system" chose "FAT" (there should also be the 
options "FAT32" and "NTFS"), press "OK" and your card will be formatted. You can use JP 
Module to format your MMC/SD card also. 
 
LIABILITY WARNING 
 

This device should be used only for experimental purposes. It has NOT gone through 
extensive testing and it could erase or corrupt some or all data on media cards that are inside the 
device. You assume to take your own risk when you purchase this device, and release the 
responsibility and liability from the manufacturer with no harm. 
 
REGULATORY WARNING 
 

This device is intended solely for experimental purpose, it is not in finished product form 
and is NOT FCC approved. If you wish to install these modules into non-experimental final 
finished products, you will be responsible to have the modules approved by the FCC at your own 
cost. 
 

Basic Stamp ® Example Programs 
 
'========================================================= 
' File........JP Module Test.BSP 
' Purpose.....This test code for JP Module 
' Auther......Jianping Sun 
' Email.......JP@JianpingUSA.com 
' Started.....Feb 08, 2008 
' Updated.....Jul. 08, 2008 
'======================================================== 
' {$STAMP BS2e} 
' {$PBASIC 2.5} 
' 
' Stamp connect to JP Module 
' 
' BS2E      JP Module 
' ----      --------- 
' P15 ----> RT (Serial in) 
' P14 <---- TX (Option) 
' P13 <---- State (Option) 
' P12 ----> Reset (Option) 
#SELECT $STAMP 
  #CASE BS2, BS2E, BS2PE 
    T2400 CON 396 
    T4800 CON 188 



    T9600 CON 84 
    T19K2 CON 32 
  #CASE BS2SX, BS2P 
    T2400 CON 1021 
    T4800 CON 500 
    T9600 CON 240 
    T19K2 CON 110 
  #CASE BS2PX 
    T2400 CON 1646 
    T4800 CON 813 
    T9600 CON 396 
    T19K2 CON 188 
#ENDSELECT 
 
i VAR Byte 
j VAR Byte 
 
SerData VAR Byte(10) 
FileName VAR Byte(12) 
FileName(0) = "J" 
FileName(1) = "P" 
FileName(2) = "M" 
FileName(3) = "O" 
FileName(4) = "D" 
FileName(5) = "0" 
FileName(6) = "0" 
FileName(7) = "0" 
FileName(8) = "T" 
FileName(9) = "X" 
FileName(10) = "T" 
FileName(11) = "!" 
 
SOUT            PIN     15 
S_IN            PIN     14 
StatePin        PIN     13 
ResetPin        PIN     12 
LED             CON     11 
Baud            CON     T2400 
'Baud  CON T4800 
'Baud  CON T9600 
'Baud  CON T19K2 
 
PAUSE 1000 
 
Main: 
  HIGH LED              'Start test indication 
'*********************************************************** 
'             Format MMC/SD Card with FAT16 
'  Warning: Format will erase ALL data on your MMC/SD card! 
'=========================================================== 
'  GOSUB Formatting         'IF your MMC/SD Card need format 
'=========================================================== 
 
  GOSUB CreateFile 
 
  GOSUB AppendFile 
 
  GOSUB OpenFile 
 
  GOSUB DeleteFile 
 
  LOW LED               'finish test indication 
END 
 
'==================== Format MMC/SD Card =================== 
 
Formatting: 
 



  SEROUT SOUT, Baud, ["MF!"]                  'Format MMC/SD Card 
  PAUSE 50 
  SEROUT SOUT, Baud, ["JPModule!!"]           'Volume label 
  PAUSE 1000 
 
Waiting: 
  IF StatePin = 0 THEN 
    GOTO  Waiting 
  ELSE 
    'Format finish 
  ENDIF 
  PAUSE 1000 
RETURN 
 
'==================== Create 10 text files ================= 
CreateFile: 
  FOR i = 0 TO 9 
    FileName(7) = 48 + i 
    SEROUT sout, Baud, ["MC!"] 
    PAUSE 10 
    SEROUT sout, Baud, [STR FileName\12] 
    PAUSE 10 
    SEROUT sout, Baud, ["20080229080808!"] '= 2/29/2008, 08:08:08 
    PAUSE 480 
    SEROUT sout, Baud, ["This is a test file number = ", FileName(7)] 
    PAUSE 180 
    SEROUT sout, Baud, ["ABCDEFGHIJKLMNOPQRSTUVWXYZ!!"] 
    PAUSE 180 
    SEROUT sout, Baud, ["1234567890##"] 
    PAUSE 180 
  NEXT 
RETURN 
 
'================ Append new data to 10 files ================ 
AppendFile: 
  FOR i = 0 TO 9 
    FileName(7) = 48 + i 
    SEROUT sout, Baud, ["MA!"] 
    PAUSE 10 
    SEROUT sout, Baud, [STR FileName\12] 
    PAUSE 10 
    SEROUT sout, Baud, ["20080229181808!"] '= 2/29/2008, 18:18:18 
    PAUSE 200 
    SEROUT sout, Baud, ["Append new data to file number = ", FileName(7)] 
    PAUSE 180 
    SEROUT sout, Baud, ["abcdefghijklmnopqrstuvwxyz!!"] 
    PAUSE 180 
    SEROUT sout, Baud, ["1234567890##"] 
    PAUSE 180 
  NEXT 
RETURN 
 
'========== Open 4 files and clear them, write new data ======== 
OpenFile: 
  FOR i = 0 TO 4 
    FileName(7) = 48 + i 
    SEROUT sout, Baud, ["MW!"] 
    PAUSE 10 
    SEROUT sout, Baud, [STR FileName\12] 
    PAUSE 10 
    SEROUT sout, Baud, ["20080229181808!"]  '= 2/29/2008, 18:18:18 
    PAUSE 200 
    SEROUT sout, Baud, ["Write new data to file number = ", FileName(7)] 
    PAUSE 180 
    SEROUT sout, Baud, ["abcdefghijklmnopqrstuvwxyz!!"] 
    PAUSE 180 
    SEROUT sout, Baud, ["1234567890##"] 
    PAUSE 180 



  NEXT 
RETURN 
 
'==================== Delete #4 file ========================= 
DeleteFile: 
  FileName(7) = 48 + 4 
  SEROUT sout, Baud, ["MD!"] 
  PAUSE 10 
  SEROUT sout, Baud, [STR FileName\12] 
  PAUSE 10 
  SEROUT sout, Baud, ["20080229181808!"] '= 2/29/2008, 18:18:18 
  PAUSE 150 
RETURN 

